Electronic Supplementary Information

(3, 2)D ^{1}H, ^{13}C BIRD^{r,X}-HSQC-TOCSY for NMR Structure Elucidation of Mixtures. Application to Complex Carbohydrates

Natalia Brodaczewska[a], Zuzana Košťálová[a,b], Dušan Uhrín[a]*

[a]EastChem School of Chemistry, University of Edinburgh Joseph Black Building, David Brewster Rd, Edinburgh, EH9 3FJ (UK),[b]Institute of Chemistry, Slovak Academy of Sciences, Dúbravská cesta 9, 845 38, Bratislava, Slovak Republic

Content

Protocol of *β*-eliminative depolymerisation of fCS……...2

Pulse sequence of the 2D ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC……..4

Pulse sequence of the 2D ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC-TOCSY……...9

AU programme for producing two 2D ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC-(TOCSY) spectra…....14

Figure S1: 1D ^{1}H and partial 2D ^{1}H, ^{13}C HSQC spectra of the fCS mixture……..16

Figure S2: Overlay of 2D ^{1}H, ^{13}C HSQC and ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC spectra of the fCS mixture……..17

Figure S3: Overlay of the 2D ^{1}H, ^{13}C HSQC-TOCSY and ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC-TOCSY spectra of the fCS mixture…….....18

Figure S4: Overlay of the methyl region of the 2D ^{1}H, ^{13}C HSQC-TOCSY and ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC-TOCSY spectra of the fCS mixture……...19

Protocol for *β*-eliminative depolymerisation of fCS

Reagents:

- benzethonium chloride (Hy^+Cl^-), 97% [Acros 121-54-0]
- N,N-Dimethyl Formamide (DMF), anhydrous [Sigma 68-12-2]
- benzyl chloride [Acros 100-44-7]
- sodium acetate, anhydrous [Fisher Scientific 127-09-3]
- sodium ethoxide (EtONa), 95 % [Sigma 141-52-6]
- sodium hydroxide (NaOH), pellets [Fisher Scientific1310-73-2]
- hydrochloric acid (HCl), ~37% [Fisher Scientific]
- ethanol absolute [VWR chemicals 64-17-5]
- de-ionized water (dH_2O)
- deuterium oxide (D_2O)

Samples:

fucosylated chondroitin sulfate (fCS) from *Holothuria forskalias* (M_w 155kDa)

Solution preparation:

<u>0.08M EtONa in ethanol:</u> 10.8 mg of EtONa to 2ml of EtOH

Procedure

<u>I. Transalification to fCS benzethonium (Hy^+) salt</u>

1) fCS (397.9 mg) was dissolved in a falcon tube in dH_2O (13.3 ml) and stirred at room temperature until it was completely solubilised.
2) Benzethonium chloride (Hy^+Cl^-) (1.155 g) was dissolved dH_2O (15 ml) (foamy solution).
3) The benzethonium chloride solution (13.3 ml) was added to the fCS solution and stirred. White precipitant was formed.
4) Left shaking for 30 min and standing for 1h.
5) Centrifuged (2218 g for 15 min; Heraeus Biofuge prime, Heraeus Instrument)
6) The sediment was dried for 5 hours in an oven (50 °C) to obtain fCS benzethonium salt (1397.2 mg).

<u>II. Esterification to fCS benzyl ester benzethonium salt</u> (lower volume of DMF and 25% higher volume of benzyl chloride together with longer time (48 h) were used to improve the esterification reaction)

7) The fCS benzethonium salt was dissolved dimethyl formamide, DMF, (17.5 ml).
8) Benzyl chloride (3.36 ml) in solution was added, heated for 48 h in oil bath at 35°C, stirred continuously.
9) Anhydrous sodium acetate in ethanol (5 ml of 10% suspension) was added to the solution of fCS benzethonium salt in DMF and left overnight in the fridge to precipitate.
10) The next day the sample was centrifuged at 2218 g for 10 min.
11) To remove DMF, the sediment was washed three times with pure ethanol (5 ml), sample collected and dried in oven (50 °C, 5 hours).
12) The dried fCS benzyl ester (650 mg) was dissolved in D_2O and subject to ^1H NMR to confirm that esterification took place.

III. Transalification

13) The fCS benzyl ester was transalified as described in I. Briefly, the sample was dissolved in dH_2O (21.7 ml) and precipitated with 0.25M HyCl (15 ml) solution. The sample seemed precipitated, but after centrifugation the supernatant was still not transparent. Extra HyCl solution (3 ml) was added; the supernatant was transparent but colorful. Dried in the oven (50 °C) overnight. Produced fCS benzyl ester benzethonium salt was weighted (1.133 mg)

IV. β-eliminative depolymerization

14) The fCS benzyl ester benzethonium salt was dissolved in DMF (2 ml) using a magnetic stirrer (took a long time).
15) Freshly prepared 0.08 M EtONa/EtOH (2 ml) was added at room temperature and shaken for 2 hours at room temperature.
16) To precipitate, a 10% suspension of sodium acetate in ethanol (9 ml) was added to the solution, stirred and left in the cold room overnight.
17) Centrifuged (2218 g for 10 mins), washed two times with 95% ethanol, sediment collected. (The white sediment of non-soluble sodium acetate was almost gone.)
18) Dried in oven at 50°C overnight and weighted (460.2 mg).

V. Saponification

19) The sediment was saponified in 0.1 M NaOH (3 ml) at 35 °C for 30 min (in a shaking incubator).
20) The solution was neutralized with 0.1M HCl (~2 ml).
21) Added pure EtOH (10 ml) and left in a cold room overnight.
22) The sediment was collected by centrifugation (2218 g for 5 min) and dried in oven (50 °) and weighted (305.8 mg), the sample was checked by ^{1}H NMR.

The estimated M_w of the product was 12.2kDa after one round of depolymerisation. The produced partially depolymerised fCS was subjected to further two rounds of depolymerisation yielding material with M_w of 8 kDa and 4.98kDa, respectively.The amount of reagents was adjusted for the starting amount of fCS obtained in the previous depolymerisation round. The 4.98 kDa M_w sample constitutes the “fCS mixture” used in this study.

The molecular weights were determined on an HPLC equipped with a refractometer and Photo Diode Array detector using the Shodex SB806M (8 mm × 300 mm) column. Mobile phase (0.2µm filtered 50mM Tris pH7 1mM EDTA 0.9%NaCl) was used at a flow rate 0.5 mL/min. A set of heparin standards (Iduron HO04-HO20) was used to calibrate the column.

Pulse sequence of the 2D ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC

```
;(3, 2)D BIRD(r,X)-HSQC
; AVANCE III, TopSpin 3.5pl7
;based on Bruker hsqcetgpsisp2.2
;avance-version (12/01/11).
;Interleaved HSQC with additional cosine or sine modulation of 13C chemical shift
;by omega(1H)
;2D H-1/X correlation via double inept transfer
;   using sensitivity improvement
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;with decoupling during acquisition
;using trim pulses in inept transfer
;using shaped pulses for all 180degree pulses on f2 - channel
;with gradients in back-inept
;
; N. Brodaczewska, Z. Košťálová, D. Uhrín, JBNMR, submitted

;A.G. Palmer III, J. Cavanagh, P.E. Wright & M. Rance, J. Magn.
;   Reson. 93, 151-170 (1991)
;L.E. Kay, P. Keifer & T. Saarinen, J. Am. Chem. Soc. 114,
;   10663-5 (1992)
;J. Schleucher, M. Schwendinger, M. Sattler, P. Schmidt, O. Schedletzky,
;   S.J. Glaser, O.W. Sorensen & C. Griesinger, J. Biomol. NMR 4,
;   301-306 (1994)
;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=
;$RECOMMEND=y

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d4=1s/(cnst2*4)"
"d11=30m"
"d12=20u"

"d0=3u"
"d25=3u"

"in0=inf1/2"
"in25=cnst1*in0"

"l0=0"

"DELTA=p16+d16+p2+d0*2-4u"
"DELTA1=p16+d16-p1*0.78+de"
;"DELTA1=p16+d16-p1*0.78+de+4u"
"DELTA2=d4-larger(p2,p14)/2-4u"
"TAU1=d4-p16-d16-larger(p2,p14)/2-4u"
```

```
"TAU2=d4+p16+d16-larger(p2,p14)/2-4u+2*d25"
"DELTA3=d24-cnst17*p24/2-p19-d16-4u"
"DELTA4=d4-larger(p2,p14)/2-p16-d16-4u"
"DELTA5=d4*2-larger(p2,p14)/2-4u"


"acqt0=0"
baseopt_echo

1 ze
  d11 pl12:f2
2 d11 do:f2

# ifdef PRESAT
  4u BLKGRAD
  d12 pl9:f1
  d1 cw:f1 ph29
  d12 do:f1
  d12 pl1:f1
  50u UNBLKGRAD
# else
  4u BLKGRAD
  d1
  50u UNBLKGRAD
# endif


if "l0 %2 == 0"
     {
   (p1 ph2)       ; sin
     }
  else
     {
   (p1 ph1)       ; cos
     }

  d25
  p16:gp5
  d16

  (p1 ph1)      ;BIRD(r,X)
  DELTA5
  4u
  (center (p2 ph2) (p14:sp3 ph1):f2 )
  4u
  DELTA5
  (p1 ph1)

  p16:gp5*-1
  d16
  d25

  TAU1 pl0:f2
  4u
  (center (p2 ph1) (p14:sp3 ph6):f2 )
```

```
 4u
 TAU2 pl2:f2
 p28 ph1
 4u
 (p1 ph2) (p3 ph3):f2 ; transfer to 13C
 d0
 (p2 ph7)
 d0
 p16:gp1*EA
 d16 pl0:f2
 (p24:sp7 ph8:r):f2
 4u
 DELTA pl2:f2
 (center (p1 ph1) (p3 ph4):f2 )
 4u
 p19:gp3
 d16
 DELTA3 pl0:f2
 (center (p2 ph1) (p24:sp7 ph9:r):f2 )
 4u
 DELTA3 pl2:f2
 p19:gp3
 d16
 (center (p1 ph2) (p3 ph5):f2 )
 4u
 p16:gp4
 d16
 DELTA4 pl0:f2
 (center (p2 ph1) (p14:sp3 ph1):f2 )
 4u
 DELTA4
 p16:gp4
 d16
 (p1 ph1)
 DELTA1
 (p2 ph1)
 4u
 p16:gp2
 d16 pl12:f2
; 4u BLKGRAD
 go=2 ph31 cpd2:f2
 d11 do:f2 mc #0 to 2
    F1I(iu0, 2)
    F1EA(calgrad(EA) & calph(ph5, +180), caldel(d0, +in0) & caldel(d25, +in25) & calph(ph3,
+180) & calph(ph6, +180) & calph(ph31, +180))
 4u BLKGRAD
exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 2 2
ph5=1 1 3 3
ph6=0
```

```
ph7=0 0 2 2
ph8=0 0 2 2
ph9=0
ph29=0
ph31=0 2 2 0

;pl0 : 0W
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl3 : f3 channel - power level for pulse (default)
;pl12: f2 channel - power level for CPD/BB decoupling
;sp3: f2 channel - shaped pulse (180degree inversion)
;spnam3: Crp60,0.5,20.1
;sp7: f2 channel - shaped pulse (180degree refocussing)
;spnam7: Crp60comp.4
;p1 : f1 channel -  90 degree high power pulse
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;     = 500usec for Crp60,0.5,20.1
;p16: homospoil/gradient pulse                      [1 msec]
;p19: gradient pulse 2                             [500 usec]
;p22: f3 channel - 180 degree high power pulse
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;p28: f1 channel - trim pulse
;d0 : incremented delay (2D)                        [3 usec]
;d25 : incremented omega(1H) evolution delay  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d4 : 1/(4J)XH
;d11: delay for disk I/O                           [30 msec]
;d16: delay for homospoil/gradient recovery
;d24: 1/(8J)XH for all multiplicities
;     1/(4J)XH for XH
;cnst2: = J(XH)
;cnst17: = -0.5 for Crp60comp.4
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;in25 = cnst1*in0
;nd0: 2
;ns: 1 * n
;ds: >= 16
;td1: twice the usual number of experiments
;FnMODE: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence


;use gradient ratio:  gp 1 : gp 2 : gp 3 : gp 4
;                     80 : 20.1 :   11 :   -5    for C-13
;                     80 :  8.1 :   11 :   -5    for N-15

;for z-only gradients:
;gpz1: 80%
;gpz2: 20.1% for C-13, 8.1% for N-15
```

```
;gpz3: 11%
;gpz4: -5%
;gpz5: 7%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam3: SMSQ10.100
;gpnam4: SMSQ10.100
;gpnam5: SMSQ10.100


;cnst17: Factor to compensate for coupling evolution during a pulse
;       (usually +1). A positive factor indicates that coupling
;       evolution continues during the pulse, whereas a negative
;       factor is necessary if the coupling is (partially) refocussed.
```

Pulse sequence of the 2D ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC-TOCSY

```
;(3, 2)D BIRD(r,X)-HSQC-TOCSY
; AVANCE III, TopSpin 3.5pl7
;Based on Bruker hsqcdietgpsisp.2
;avance-version (12/01/11)
;Interleaved HSQC-TOCSY with additional cosine or sine modulation of 13C chemical shift
;by omega(1H)
;2D H-1/X correlation via double inept transfer
;   using sensitivity improvement and DIPSI2
;   for homonuclear Hartman-Hahn mixing
;phase sensitive using Echo/Antiecho-TPPI gradient selection
;with decoupling during acquisition - using f2
;using trim pulses in inept transfer
;using shaped pulses for all 180degree pulses on f2 – channel

; N. Brodaczewska, Z. Košťálová, D. Uhrín, JBNMR, submitted

;
;$CLASS=HighRes
;$DIM=2D
;$TYPE=
;$SUBTYPE=
;$COMMENT=
;$RECOMMEND=y

#include <Avance.incl>
#include <Grad.incl>
#include <Delay.incl>

"p2=p1*2"
"d4=1s/(cnst2*4)"
"d11=30m"
"d12=20u"

"d0=3u"
"d25=3u"

"in0=inf1/2"
"in25=cnst1*in0"

"l0=0"

"FACTOR1=(d9/(p6*115.112))/2"
"l1=FACTOR1*2"
"d12=20u"

"DELTA=p16+d16+p2+d0*2-4u"
"DELTA1=p16+d16-p1*0.78+de"
;"DELTA1=p16+d16-p1*0.78+de+4u"
"DELTA2=d4-larger(p2,p14)/2-4u"
"TAU1=d4-p16-d16-larger(p2,p14)/2-4u"
"TAU2=d4+p16+d16-larger(p2,p14)/2-4u+2*d25"
"DELTA3=d24-cnst17*p24/2-4u"
"DELTA4=d4-larger(p2,p14)/2-4u-2*p3-6u-p16-d16"
```

```
"DELTA5=d4*2-larger(p2,p14)/2-4u"
"DELTA6=d4-larger(p2,p14)/2-4u-p1*0.78+de"


"acqt0=0"
baseopt_echo

1 ze
  d11 pl12:f2
2 d11 do:f2

#ifdef PRESAT
  4u BLKGRAD
  d12 pl9:f1
  d1 cw:f1 ph29
  d12 do:f1
  d12 pl1:f1
  50u UNBLKGRAD
#else
  4u BLKGRAD
  d1
  50u UNBLKGRAD
#endif


if "l0 %2 == 0"
     {
   (p1 ph2)     ; sin
     }
  else
     {
   (p1 ph1)     ; cos
     }

  d25
  p16:gp5
  d16 pl0:f2

  (p1 ph1)      ;BIRD(r,X)
  DELTA5
  4u
  (center (p2 ph2) (p14:sp3 ph1):f2 )
  4u
  DELTA5
  (p1 ph1)

  p16:gp5*-1
  d16
  d25

  TAU1
  4u
  (center (p2 ph1) (p14:sp3 ph6):f2 )
  4u
  TAU2 pl2:f2
```

```
  p28 ph1
  4u
  (p1 ph2)

  (p3 ph3):f2  ; transfer to 13C
  d0
  (p2 ph7)
  d0
  p16:gp1*EA
  d16 pl0:f2
  (p24:sp7 ph4):f2
  4u
  DELTA pl2:f2
  (center (p1 ph1) (p3 ph4):f2 )
  4u
  DELTA3 pl0:f2
  (center (p2 ph1) (p24:sp7 ph1):f2 )
  4u
  DELTA3 pl2:f2
  (center (p1 ph2) (p3 ph5):f2 )
  4u
  DELTA2 pl0:f2
  (center (p2 ph1) (p14:sp3 ph1):f2 )
  4u
  DELTA2 pl10:f1

                                                    ;begin DIPSI2
4 p6*3.556 ph22
  p6*4.556 ph24
  p6*3.222 ph22
  p6*3.167 ph24
  p6*0.333 ph22
  p6*2.722 ph24
  p6*4.167 ph22
  p6*2.944 ph24
  p6*4.111 ph22

  p6*3.556 ph24
  p6*4.556 ph22
  p6*3.222 ph24
  p6*3.167 ph22
  p6*0.333 ph24
  p6*2.722 ph22
  p6*4.167 ph24
  p6*2.944 ph22
  p6*4.111 ph24

  p6*3.556 ph24
  p6*4.556 ph22
  p6*3.222 ph24
  p6*3.167 ph22
  p6*0.333 ph24
  p6*2.722 ph22
  p6*4.167 ph24
  p6*2.944 ph22
```

```
  p6*4.111 ph24

  p6*3.556 ph22
  p6*4.556 ph24
  p6*3.222 ph22
  p6*3.167 ph24
  p6*0.333 ph22
  p6*2.722 ph24
  p6*4.167 ph22
  p6*2.944 ph24
  p6*4.111 ph22
  lo to 4 times l1
                                                ;end DIPSI2

  4u pl1:f1
  (p1 ph1)

  DELTA1
  (p2 ph1)
  4u
  p16:gp2
  d16 pl12:f2
; 4u BLKGRAD

  go=2 ph31 cpd2:f2
  d11 do:f2 mc #0 to 2
     F1I(iu0, 2)
     F1EA(calgrad(EA) & calph(ph5, +180), caldel(d0, +in0) & caldel(d25, +in25) & calph(ph3,
+180) & calph(ph6, +180) & calph(ph31, +180))
exit


ph1=0
ph2=1
ph3=0 2
ph4=0 0 2 2
ph5=1 1 3 3
ph6=0
ph7=0 0 2 2
ph22=3
ph24=1
ph29=0
ph31=0 2 2 0


;pl0 : 0W
;pl1 : f1 channel - power level for pulse (default)
;pl2 : f2 channel - power level for pulse (default)
;pl10: f1 channel - power level for TOCSY-spinlock
;pl12: f2 channel - power level for CPD/BB decoupling
;sp3: f2 channel - shaped pulse (180degree inversion)
;spnam3: Crp60,0.5,20.1
;sp7: f2 channel - shaped pulse (180degree refocussing)
;spnam7: Crp60comp.4
;p1 : f1 channel -  90 degree high power pulse
```

```
;p2 : f1 channel - 180 degree high power pulse
;p3 : f2 channel -  90 degree high power pulse
;p6 : f1 channel -  90 degree low power pulse
;p14: f2 channel - 180 degree shaped pulse for inversion
;     = 500usec for Crp60,0.5,20.1
;p16: homospoil/gradient pulse                       [1 msec]
;p24: f2 channel - 180 degree shaped pulse for refocussing
;     = 2msec for Crp60comp.4
;p28: f1 channel - trim pulse                        [1 msec]
;d0 : incremented delay (2D)                         [3 usec]
;d25 : incremented omega(1H) evolution delay  [3 usec]
;d1 : relaxation delay; 1-5 * T1
;d4 : 1/(4J(XH))
;d9 : TOCSY mixing time
;d11: delay for disk I/O                             [30 msec]
;d16: delay for homospoil/gradient recovery
;d24: 1/(8J)XH for all multiplicities
;     1/(4J)XH for XH
;cnst1: scaling factor for omega(1H) relative to omega(13C)
;cnst2: = J(XH)
;cnst17: = -0.5 for Crp60comp.4
;l1: loop for DIPSI cycle: ((p6*115.112) * l1) = mixing time
;inf1: 1/SW(X) = 2 * DW(X)
;in0: 1/(2 * SW(X)) = DW(X)
;in25 = cnst1*in0
;nd0: 2
;ns: 4 * n
;ds: >= 16
;td1: twice the usual number of experiments
;FnMODE: echo-antiecho
;cpd2: decoupling according to sequence defined by cpdprg2
;pcpd2: f2 channel - 90 degree pulse for decoupling sequence

;use gradient ratio:      gp 1 : gp 2
;                           80 : 20.1       for C-13
;                           80 :  8.1       for N-15

;for z-only gradients:
;gpz1: 80%
;gpz2: 20.1% for C-13, 8.1% for N-15
;gpz5: 7%

;use gradient files:
;gpnam1: SMSQ10.100
;gpnam2: SMSQ10.100
;gpnam5: SMSQ10.100

;cnst17: Factor to compensate for coupling evolution during a pulse
;       (usually +1). A positive factor indicates that coupling
;       evolution continues during the pulse, whereas a negative
;       factor is necessary if the coupling is (partially) refocussed.
```

AU programme for producing two complementary 2D ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC-(TOCSY) spectra.

```
/*****************************************************************/
/*    NAME            17/10/2017           */
/*****************************************************************/
/*    Short Description :                          */
/*    Processes interleaved (3, 2)D BIRD(r,X)-HSQC-(TOCSY) dataset
      */
/*****************************************************************/
/*    Keywords :                           */
/*    split, add2d, interleaves                                  */
/*****************************************************************/
/*    Description/Usage :                          */
/*    Splits an interleaved experiment into two halves */
/*    Prompts the user for a new expno for the two halves
      */
/*    Adds a 90 degree PHC0 correction to the first half
      */
/*    Saves second half to procno 3                */
/*    Adds both halves with alpha/gamma =1         */
/*    Saves this procno 2         */
/*    Adds both halves with alpha=1,gamma=-1       */
/************************************************/
/*****************************************************************/
/*    Author(s) :                                  */
/*    Name       : Will Kew            */
/*    Organisation     : University of Edinburgh              */
/*    Email      : w.kew@sms.ed.ac.uk         */
/*    Email      : will.kew@gmail.com         */
/*****************************************************************/
/*    Name       Date  Modification:               */
/*    wrk        17/10/2017 created                      */
/*****************************************************************/
/*
$Id: Name,v 1.0 17/10/2017 10:11:14 wrk Exp $
*/

#include <inc/sysutil>
#include <inc/exptUtil>
#include <CUtil/CUtil.h>

int     first,iexpno;
float f1phc0;
//double  swp;
char buffer[1024]; //this buffer is for the XCMD string for
splitting, may need memory optimisation

//get current data
GETCURDATA

//define our new expnos for the split data
first=expno;
iexpno = expno+10000;
```

```
GETINT("Enter the EXPNO for the split data? i.e. 10000+current
expno", iexpno)
if (first==iexpno)
     STOPMSG("program aborted\nYou don't want to split into the
original dataset")

//generate and execute the split command. Calls the default bruker
AU program "split".
snprintf(buffer, sizeof(buffer), "split 2 %d", iexpno);
XCMD(buffer);

DATASET(name, iexpno, 1, disk, user)
VIEWDATA_SAMEWIN


FETCHPAR1("PHC0",&f1phc0);
double newf1pch0 = f1phc0+90.0; //90degree phase correction on the
first half of the data
STOREPAR1("PHC0",newf1pch0); //stores new F1 phase correction
XFB;


DATASET(name, iexpno+1, procno, disk, user)
VIEWDATA_SAMEWIN
XFB;
WRP(3);

STOREPAR("ALPHA",1.0);
STOREPAR("GAMMA",1.0);


// define second dataset
DATASET2(name, iexpno, procno, disk, user);
ADD2D;
WRP(2);

DATASET(name, iexpno+1, procno, disk, user)
VIEWDATA_SAMEWIN
XFB;

STOREPAR("ALPHA",1.0);
STOREPAR("GAMMA",-1.0);

DATASET2(name, iexpno, procno, disk, user);
ADD2D;

DATASET(name, iexpno+1, 1, disk, user)
VIEWDATA_SAMEWIN

snprintf(buffer, sizeof(buffer), "Processing complete\nSee expno %d
and procnos 1,2 for results", iexpno+1);
QUITMSG(buffer);
```



Figure S1. (a) 1D 1H NMR spectrum of the fCS mixture. Impurities are marked by an asterisk. (b) Non-anomeric region of the 2D ^{1}H, ^{13}C HSQC spectrum of the fCS mixture showing large degeneracy of ^{13}C chemical shifts. 1D ^{1}H spectrum and the positive projection of the 2D spectrum are shown at the top and the side, respectively.



Figure S2: Overlay of 2D ^{1}H, ^{13}C HSQC spectrum (red) and cosine-modulated ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC spectrum (blue) of the fCS mixture. 1D ^{1}H spectrum and the positive projection of the 2D spectrum are shown at the top and the side, respectively. For parameters see Materials and methods.



Figure S3: A 2D ^{1}H, ^{13}C HSQC-TOCSY spectrum (red) and a cosine-modulated ^{1}H, ^{13}C (3, 2)D $BIRD^{r,X}$-HSQC-TOCSY spectrum (blue) of the fCS mixture. 1D ^{1}H spectrum and the positive projection of the 2D spectrum are shown at the top and the side, respectively. The methyl region is shown in Figure S4. For parameters Materials and methods.



Figure S4: (a) The methyl region of 2D ^{1}H, ^{13}C HSQC-TOCSY spectrum (violet) and ^{1}H, ^{13}C (3, 2)D BIRD^{r,X}-HSQC-TOCSY spectra (blue and red) of the fCS mixture. (b) expansions of the regions indicated in (a). Dashed horizontal lines highlight the achieved separation of cross peaks. A more substantial separation is seen in the Ω_{13C} - Ω_{1H} spectrum (A).